﻿CREATE PROCEDURE [Event].[DeleteEvent]
	@EventID INT,
	@OwnerID INT
WITH EXECUTE AS OWNER
AS
SET NOCOUNT ON;

BEGIN TRY
  BEGIN TRANSACTION
  
  DELETE es
    FROM Event.Schedule AS es
         INNER JOIN Event.Events AS ee
            ON es.EventID = ee.EventID
   WHERE ee.EventID = @EventID
     AND ee.OwnerID = @OwnerID;
  
  DELETE Event.Events
   WHERE EventID = @EventID
     AND OwnerID = @OwnerID;
  
  COMMIT TRANSACTION
END TRY
BEGIN CATCH
  IF @@TRANCOUNT > 0
    ROLLBACK TRANSACTION;
  
  DECLARE @ERROR_SEVERITY AS INT,
          @ERROR_STATE AS INT,
          @ERROR_NUMBER AS INT,
          @ERROR_LINE AS INT,
          @ERROR_MESSAGE AS NVARCHAR(4000);

  SELECT @ERROR_SEVERITY = ERROR_SEVERITY(),
         @ERROR_STATE = ERROR_STATE(),
         @ERROR_NUMBER = ERROR_NUMBER(),
         @ERROR_LINE = ERROR_LINE(),
         @ERROR_MESSAGE = ERROR_MESSAGE();

  RAISERROR('Msg %d, Line %d, :%s',
            @ERROR_SEVERITY,
            @ERROR_STATE,
            @ERROR_NUMBER,
            @ERROR_LINE,
            @ERROR_MESSAGE);
END CATCH